Multiple Fourier Series

William O. Bray

Herein we will explore a few examples of multiple Fourier sine series on the square $D = [0, \pi] \times [0, \pi]$. These have the form:

$$f(x, y) \sim \sum_{n,m} \hat{f}(n, m) \sin(n x) \sin(n y), \qquad \hat{f}(n, m) = \frac{4}{\pi^2} \int_D f(x, y) \sin(n x) \sin(m y)\, dA_{(x,y)}$$

The following command forms the square partial sums of the Fourier sine series with Fourier coefficients given as a table.

In[23]:=

```
parsum[x_, y_, N_, a_] :=
 Sum[a[[n, m]] Sin[n x] Sin[m y], {n, 1, N}, {m, 1, N}]
```

Example 1

In[24]:=

```
f1[x_, y_] := x (Pi - x) y (Pi - y)
```

In[25]:=

```
Plot3D[f1[x, y], {x, 0, Pi}, {y, 0, Pi}, AxesLabel → {x, y, z}]
```

Out[25]=



This function is actually a product of a function of x with a function of y, hence the double Fourier sine coefficients are the products of the one dimensional Fourier coefficients. We can implement this as follows.

In[26]:=
```
g[k_] := FourierSinCoefficient[x (Pi - x), x, k]
```

In[27]:=
```
f1hat[n_, m_] := g[n] g[m]
```

In[28]:=
```
f1hat[n, m]
```

Out[28]=

$$\frac{16 \left(-1+(-1)^m\right)\left(-1+(-1)^n\right)}{m^3\, n^3\, \pi^2}$$

In[29]:=
```
a1 = Table[f1hat[n, m], {n, 1, 10}, {m, 1, 10}];
```

In[30]:=
```
Plot3D[parsum[x, y, 10, a1], {x, 0, Pi}, {y, 0, Pi}]
```

Out[30]=



Visually, we cannot see much difference between the partial sum plot and that of the actual function. The following is a depiction of the error.

In[31]:=
```
Plot3D[Abs[f1[x, y] - parsum[x, y, 10, a1]],
 {x, 0, Pi}, {y, 0, Pi}, PlotPoints → 40]
```

Out[31]=



Notice the growth in error along the boundaries; undoubtably this is due to the fact that the partial derivatives of the function are discontinuous across the boundary (think in terms of the periodic extension of this function).

Example 2

In[32]:=
```
f2[x_, y_] := Piecewise[{{x y, 0 ≤ x ≤ Pi / 2 && 0 ≤ y ≤ Pi / 2},
   {x (Pi - y), 0 ≤ x < Pi / 2 && Pi / 2 ≤ y ≤ Pi},
   {(Pi - x) y, Pi / 2 ≤ x ≤ Pi && 0 ≤ y ≤ Pi / 2},
   {(Pi - x) (Pi - y), Pi / 2 ≤ x ≤ Pi && Pi / 2 ≤ y ≤ Pi}}]
```

In[33]:=
```
Plot3D[f2[x, y], {x, 0, Pi}, {y, 0, Pi}]
```

Out[33]=



In[34]:=
```
Integrate[f2[x, y] Sin[n x] Sin[m y], {x, 0, Pi}, {y, 0, Pi}]
```

Out[34]=

$$\frac{1}{m^2 n^2}\left(4 \operatorname{Sin}\left[\frac{m \pi}{2}\right] \operatorname{Sin}\left[\frac{n \pi}{2}\right] - 2 \operatorname{Sin}[m \pi] \operatorname{Sin}\left[\frac{n \pi}{2}\right] - 2 \operatorname{Sin}\left[\frac{m \pi}{2}\right] \operatorname{Sin}[n \pi] + \operatorname{Sin}[m \pi] \operatorname{Sin}[n \pi]\right)$$

In[35]:=
```
a2 = (4 / Pi^2) Table[%, {n, 1, 10}, {m, 1, 10}];
```

In[36]:=

```
Plot3D[parsum[x, y, 10, a2], {x, 0, Pi}, {y, 0, Pi}]
```

Out[36]=



For a look at the error, consider the following.

In[37]:=

```
Plot3D[Abs[f2[x, y] - parsum[x, y, 10, a2]],
 {x, 0, Pi}, {y, 0, Pi}, PlotRange → Full, PlotPoints → 25]
```

Out[37]=



Example 3

In[38]:=

```
f3[x_, y_] :=
 Piecewise[{{1, 0 ≤ x ≤ Pi && 0 ≤ y ≤ x}, {0, 0 ≤ x ≤ Pi && x < y ≤ Pi}}]
```

In[39]:=

```
Plot3D[f3[x, y], {x, 0, Pi}, {y, 0, Pi}]
```

Out[39]=



In[40]:=

```
Integrate[Sin[n x] Sin[m y], {x, 0, Pi}, {y, 0, x}]
```

Out[40]=

$$\frac{1 - \text{Cos}[n\,\pi]}{m\,n} + \frac{n - n\,\text{Cos}[m\,\pi]\,\text{Cos}[n\,\pi] - m\,\text{Sin}[m\,\pi]\,\text{Sin}[n\,\pi]}{m^3 - m\,n^2}$$

In[41]:=

```
a3 =
  Table[(4 / Pi^2) Integrate[Sin[n x] Sin[m y], {x, 0, Pi}, {y, 0, x}],
   {n, 1, 15}, {m, 1, 15}];
```

In[42]:=

```
Plot3D[parsum[x, y, 5, a3], {x, 0, Pi}, {y, 0, Pi}]
```

Out[42]=



In[43]:=

```
Plot3D[parsum[x, y, 10, a3], {x, 0, Pi}, {y, 0, Pi}]
```

Out[43]=



In[44]:=
```
Plot3D[parsum[x, y, 15, a3], {x, 0, Pi}, {y, 0, Pi}, PlotPoints → 30]
```

Out[44]=



A view of the partial sums relative to the periodic extension of this function is seen as follows.

In[45]:=
```
Plot3D[parsum[x, y, 15, a3], {x, 0, 2 Pi}, {y, 0, 2 Pi}]
```

Out[45]=